Math 132
Fall 2015 Exam 2

Formulas

$\pi = 3.141592654$, $\ln(1) = 0$, $\ln(e) = 1$, $\ln(x\,y) = \ln(x) + \ln(y)$, $\ln(x^p) = p\ln(x)$

$\sin\left(\frac{\pi}{6}\right) = \cos\left(\frac{\pi}{3}\right) = \frac{1}{2}$, $\sin\left(\frac{\pi}{3}\right) = \cos\left(\frac{\pi}{6}\right) = \frac{\sqrt{3}}{2}$, $\sin\left(\frac{\pi}{4}\right) = \cos\left(\frac{\pi}{4}\right) = \frac{\sqrt{2}}{2}$,

$\sin(0) = \sin(\pi) = \cos\left(\frac{\pi}{2}\right) = 0$, $\sin\left(\frac{\pi}{2}\right) = \cos(0) = \tan\left(\frac{\pi}{4}\right) = 1$, $\sin\left(\frac{3\pi}{2}\right) = \cos(\pi) = -1$

$\sin(2x) = 2\sin(x)\cos(x)$, $2\cos(x)^2 = 1 + \cos(2x)$, $2\sin(x)^2 = 1 - \cos(2x)$,

$\int \frac{1}{x}\,dx = \ln(|x|) + C$, $\int \ln(x)\,dx = x\ln(x) - x + C$, $\int a^x\,dx = \frac{a^x}{\ln(a)} + C$,

$\int u\,dv = u\,v - \int v\,du$

$\int \frac{1}{a^2 + x^2}\,dx = \frac{1}{a}\arctan\left(\frac{x}{a}\right) + C$, $\int \frac{1}{\sqrt{a^2 - x^2}}\,dx = \arcsin\left(\frac{x}{a}\right) + C$,

$\int \frac{1}{|x|\sqrt{x^2 - a^2}}\,dx = \frac{1}{a}\operatorname{arcsec}\left(\frac{x}{a}\right) + C$, $1 - \sin(x)^2 = \cos(x)^2$, $1 + \tan(x)^2 = \sec(x)^2$

$\int \sec(x)^2\,dx = \tan(x) + C$, $\int \csc(x)^2\,dx = -\cot(x) + C$,

$\int 2\sec(x)^3\,dx = \sec(x)\tan(x) + \ln(|\sec(x) + \tan(x)|) + C$

$\int \sec(x)\tan(x)\,dx = \sec(x) + C$, $\int \csc(x)\cot(x)\,dx = -\csc(x) + C$

$\int \tan(x)\,dx = \ln(|\sec(x)|) + C$, $\int \sec(x)\,dx = \ln(|\sec(x) + \tan(x)|) + C$

$$\int \cos(x)^n\,dx = \frac{1}{n}\sin(x)\cos(x)^{(n-1)} + \frac{n-1}{n}\int \cos(x)^{(n-2)}\,dx$$

1. A continuous function f is defined on the interval I = [2 , 14]. The average value of f on I is 3. The average value of f on the subinterval [2 , 6] is 7. What is $\int_6^{14} f(x)\,dx$?

a) 5 b) 6 c) 7 d) 8 e) 9

f) 10 g) 11 h) 12 i) 13 j) 14

Solution: d

Let A, B, C, denote the integrals over [2 , 6], [6 , 14], and [2 , 14] respectively. Then

```
> A := ( 6 - 2)*7;
  C := (14 - 2)*3;
```

$$A := 28$$

$$C := 36$$

```
> B = C - A;
```

$$B = 8$$

2. A differentiable function g has the values $g(0) = 1$, $g(2) = 4$, and $g(4) = 16$. Estimate the integral $\int_0^4 x\, D(g)(x)\, dx$ by integrating by parts and then applying Simpson's Rule to the resulting integral.

a) 40 b) 41 c) 42 d) 43 e) 44
f) 45 g) 46 h) 47 i) 48 j) 49

Solution: c

```
> g(0) := 1;  g(2) := 4;  g(4) := 16;
```

$$g(0) := 1$$

$$g(2) := 4$$

$$g(4) := 16$$

```
> Delta_x := (4-0)/2;
```

$$Delta\_x := 2$$

```
> value( subs(x = 4, x*g(x)) - subs(x = 0, x*g(x)) -
  (Delta_x/3)*(1*g(0) + 4*g(2) + 1*g(4)) );
```

$$42$$

The problem is over, but it may be of interest to know that the model for the partially-defined function g in this problem was
$g(x) = 2^x$, It turns out that the estimate of 42 for the model function's integral is reasonably accurate given that the number of subintervals is so small.

```
> g := x -> 2^x;
```

$$g := x \rightarrow 2^x$$

```
> g(0), g(2), g(4);
```

$$1, 4, 16$$

```
> eqn := Int(x*'D(g)(x)',x=0..4) = int(x*'D(g)(x)',x=0..4);
```

$$eqn := \int_0^4 x\,\mathrm{D}(g)(x)\,dx = \frac{64\ln(2) - 15}{\ln(2)}$$

```
> evalf(rhs(eqn));
```

$$42.35957439$$

3. Calculate $\displaystyle\int_0^{\frac{\pi}{4}} 8\,x\cos(2\,x)\,dx$ **.**

a) $\pi - 3$ **b)** $\pi + 3$ **c)** $\pi - 2$ **d)** $\pi + 2$ **e)** π
f) $\pi - 1$ **g)** $\pi + 1$ **h)** $2\,\pi - 1$ **i)** $2\,\pi + 1$ **j)** $2\,\pi$

Solution: c

```
> eqn1 := J = Int(8*x*cos(2*x),x = 0 .. Pi/4);
  #
  # Call the value we seek J for easy reference
```

$$eqn1 := J = \int_0^{\frac{\pi}{4}} 8\,x\cos(2\,x)\,dx$$

```
> eqn2 := J = intparts(rhs(eqn1), x);
  #
  # Here we integrate by parts with u = x, dv = 8cos(2x)dx.
```

$$eqn2 := J = \pi - \int_0^{\frac{\pi}{4}} 4\sin(2\,x)\,dx$$

```
> eqn3 := lhs(eqn2) = value(rhs(eqn2));
  #
  # Here we mop up, evaluating int(sin(2*x), x = 0 .. P/4)
```

$$eqn3 := J = \pi - 2$$

As a one-line verification, we will use Maple's built-in integrator:

```
> rhs(eqn1) = value(rhs(eqn1));
```

$$\int_0^{\frac{\pi}{4}} 8\,x\cos(2\,x)\,dx = \pi - 2$$

4. Let $F(x)$ be the antiderivative of $x^2\,e^x$ with the specific constant of integration that results in the equation $F(0) = 2$. What is the product $e\,F(-1)$?

a) −5 b) −4 c) −3 d) −2 e) −1
f) 1 g) 2 h) 3 i) 4 j) 5

Solution: j

```
> J := Int(x^2*exp(x), x);
```

$$J := \int x^2\,e^x\,dx$$

```
> J = intparts( Int(x^2*exp(x),x), x^2);
  #
  # Integrate by parts with u = x^2, dv = exp(x)dx, du = 2x dx,
  v = exp(x)
```

$$\int x^2\,e^x\,dx = x^2\,e^x - \int 2\,x\,e^x\,dx$$

```
> J = x^2*exp(x) - 2*intparts(Int(x*exp(x),x),  x);
  #
  # Integrate by parts with u = x, dv = exp(x)dx, du = dx, v =
  exp(x)
```

$$\int x^2\,e^x\,dx = x^2\,e^x - 2\,x\,e^x + 2\int e^x\,dx$$

```
> J = x^2*exp(x) - 2*x*exp(x) + 2*value(Int(exp(x),x)) + C;
  #
  # Evaluate the routine integral that remains
```

$$\int x^2\,e^x\,dx = x^2\,e^x - 2\,x\,e^x + 2\,e^x + C$$

```
> J = factor(x^2*exp(x)-2*x*exp(x)+2*exp(x)) + C;
```

```
#
# Factor exp(x)
```

$$\int x^2 \mathbf{e}^x \, dx = (x^2 - 2\,x + 2)\,\mathbf{e}^x + C$$

```
> F := x -> (x^2-2*x+2)*exp(x)+C;
#
# Defines F as a function
```

$$F := x \to (x^2 - 2\,x + 2)\,\mathbf{e}^x + C$$

```
> testeq( D(F)(x) = integrand(J) );
#
# Verification by differentiation that F(x) is the correcy
antiderivative
```

$$true$$

Next we solve for the value of the constant of integration

```
> eqn := F(0) = 2;
```

$$eqn := 2 + C = 2$$

```
> C = solve(eqn, C);
```

$$C = 0$$

```
> F := x -> (x^2-2*x+2)*exp(x);
#
# We redefine F by setting C = 0
```

$$F := x \to (x^2 - 2\,x + 2)\,\mathbf{e}^x$$

```
> answer := simplify( exp(1)*F(-1) );
#
# Now we calculate e*F(-1)   (NB: e is not predefined in
Maple so we use exp(1), which is e
```

$$answer := 5$$

Maple's differential equation solver calculates F(x) more directly.

```
> eqn2 := dsolve({D(F)(x) = x^2*exp(x), F(0) = 2}, F(x));
```

$$eqn2 := \mathrm{F}(x) = x^2\,\mathbf{e}^x - 2\,x\,\mathbf{e}^x + 2\,\mathbf{e}^x$$

```
> eqn3 := subs(x = -1, map( z -> exp(1)*z, eqn2));
```

$$eqn3 := \mathbf{e}\,\mathrm{F}(-1) = 5\,\mathbf{e}\,\mathbf{e}^{(-1)}$$

```
> eqn4 := map(simplify, eqn3);
```

$$eqn4 := \mathbf{e}\,\mathrm{F}(-1) = 5$$

5. It is not difficult to show that

$$\int_1^2 4\,x^{\left(\frac{1}{2}\right)}\,(2-x)^{\left(\frac{1}{2}\right)}\,dx = \pi$$

by combining the two square roots in the integrand and then making an appropriate trigonometric substitution. Use this given evaluation to calculate

$$\int_1^2 4\,x^{\left(-\frac{1}{2}\right)}\,(2-x)^{\left(\frac{3}{2}\right)}\,dx.$$

a) $2(\pi-1)$ b) $2\pi-3$ c) $2(\pi-2)$ d) $2\pi-5$ e) $2(\pi-3)$
f) $3\pi-5$ g) $3(\pi-2)$ h) $3\pi-7$ i) $3\pi-8$ j) $3(\pi-3)$

Solution: i

```
> eqn1 := J = Int(4*x^(1/2)*(2-x)^(1/2),x = 1 .. 2);
```

$$eqn1 := J = \int_1^2 4\sqrt{x}\,\sqrt{2-x}\,dx$$

```
> value(rhs(eqn1));
  #
  # Verifying the given information
```

$$\pi$$

```
> eqn2 := K = Int(4*x^(-1/2)*(2-x)^(3/2),x = 1 .. 2);
  #
  # K is the integral that is to be calculated
```

$$eqn2 := K = \int_1^2 \frac{4\,(2-x)^{(3/2)}}{\sqrt{x}}\,dx$$

```
> eqn3 := K = intparts(rhs(eqn2), (2-x)^(3/2));
  #
  # Integration by parts with u = (2-x)^(3/2), dv =
  (4/sqrt(x))*dx
```

$$eqn3 := K = -8 - \int_1^2 -12\sqrt{x}\,\sqrt{2-x}\,dx$$

This equation tells us that

$$\int_1^2 4\,x^{\left(-\frac{1}{2}\right)}\,(2-x)^{\left(\frac{3}{2}\right)}\,dx =$$

$$K = -8 + 3\int_1^2 4\,x^{\left(\frac{1}{2}\right)}\,(2-x)^{\left(\frac{1}{2}\right)}\,dx\ .$$

or

$$\int_1^2 4\,x^{\left(-\frac{1}{2}\right)}\,(2-x)^{\left(\frac{3}{2}\right)}\,dx = -8 + 3\,\pi\,.$$

```
> Int(4*x^(-1/2)*(2-x)^(3/2), x = 1 .. 2) =
  int(4*x^(-1/2)*(2-x)^(3/2), x = 1 .. 2);
  #
  # Verification using Maple's builtin integrator
```

$$\int_1^2 \frac{4\,(2-x)^{(3/2)}}{\sqrt{x}}\,dx = 3\,\pi - 8$$

6. Evaluate $\displaystyle\int_0^{\frac{\pi}{3}} 35\,\sin(x)^3\,\cos(x)^4\,dx.$

a) $\frac{115}{64}$ **b)** $\frac{231}{128}$ **c)** $\frac{29}{16}$ **d)** $\frac{233}{128}$ **e)** $\frac{117}{64}$

f) $\frac{235}{128}$ **g)** $\frac{59}{32}$ **h)** $\frac{237}{128}$ **i)** $\frac{119}{64}$ **j)** $\frac{239}{128}$

Solution: d

Write $\sin(x)^3 = \sin(x)^2\,\sin(x)$, or $\sin(x)^3 = (1-\cos(x)^2)\,\sin(x)$ in preparation for the substitution $u = \cos(x)$, $du = -\sin(x)\,dx$.

```
> eqn1 := Int(35*sin(x)^3*cos(x)^4,x=0..Pi/3) =
  Int(35*(1-cos(x)^2)*cos(x)^4*sin(x),x=0..Pi/3) ;
  #
  # Write sin(x)^3 as sin(x)^2*sin(x), or (1 - cos(x)^2)*sin(x)
```

```
in preparation for a substitution
```

$$eqn1 := \int_0^{\frac{\pi}{3}} 35 \sin(x)^3 \cos(x)^4 \, dx = \int_0^{\frac{\pi}{3}} 35 \left(1 - \cos(x)^2\right) \cos(x)^4 \sin(x) \, dx$$

```
> eqn2 := lhs(eqn1) = changevar( u = cos(x), rhs(eqn1), u);
#
# Substitute u = cos(x), du = -sin(x)dx
```

$$eqn2 := \int_0^{\frac{\pi}{3}} 35 \sin(x)^3 \cos(x)^4 \, dx = \int_{1/2}^{1} 35 \left(1 - u^2\right) u^4 \, du$$

```
> eqn3 := lhs(eqn2) = Int(expand(integrand(rhs(eqn2))),
u=1/2..1);
#
# Expand the integrand on the right side
```

$$eqn3 := \int_0^{\frac{\pi}{3}} 35 \sin(x)^3 \cos(x)^4 \, dx = \int_{1/2}^{1} 35 u^4 - 35 u^6 \, du$$

```
> eqn4 := lhs(eqn3) = value(rhs(eqn3));
#
# Evaluate the routine integral on the right
```

$$eqn4 := \int_0^{\frac{\pi}{3}} 35 \sin(x)^3 \cos(x)^4 \, dx = \frac{233}{128}$$

Verification using maple's builtin integrator:

```
> Int(35*sin(x)^3*cos(x)^4,x=0..Pi/3) =
int(35*sin(x)^3*cos(x)^4,x=0..Pi/3);
```

$$\int_0^{\frac{\pi}{3}} 35 \sin(x)^3 \cos(x)^4 \, dx = \frac{233}{128}$$

7. Evaluate $\displaystyle \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\sec(\theta)}{\tan(\theta)^2} \, d\theta.$

a) $\frac{1}{2}\sqrt{2}$ b) $2-\sqrt{2}$ c) $1-\frac{1}{2}\sqrt{2}$ d) $1+\frac{1}{2}\sqrt{2}$ e) $\frac{1}{2}(1+\sqrt{2})$

f) $\sqrt{2}-\frac{1}{2}$ g) $\sqrt{2}-1$ h) $\sqrt{3}-\frac{3}{2}$ i) $\sqrt{3}-1$ j) $2-\sqrt{3}$

Solution: b

```
> eqn1 := Int(sec(theta)/(tan(theta)^2),theta = Pi/6 .. Pi/4) =
  Int(convert(sec(theta)/(tan(theta)^2),sincos),theta = Pi/6 ..
  Pi/4);
  #
  # Convert the trig functions to sines and cosines
```

$$eqn1 := \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\sec(\theta)}{\tan(\theta)^2}\, d\theta = \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\cos(\theta)}{\sin(\theta)^2}\, d\theta$$

```
> eqn2 := lhs(eqn1) = changevar(u = sin(theta), rhs(eqn1), u);
  #
  # Make the substitution u = sin(theta), du = cos(theta)
```

$$eqn2 := \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\sec(\theta)}{\tan(\theta)^2}\, d\theta = \int_{1/2}^{\frac{\sqrt{2}}{2}} \frac{1}{u^2}\, du$$

```
> eqn3 := lhs(eqn2) = value(rhs(eqn2));
```

$$eqn3 := \int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\sec(\theta)}{\tan(\theta)^2}\, d\theta = -\sqrt{2}+2$$

Verification using maple's builtin integrator:

```
> Int(sec(theta)/(tan(theta)^2),theta = Pi/6 .. Pi/4) =
  int(sec(theta)/(tan(theta)^2),theta = Pi/6 .. Pi/4);
```

$$\int_{\frac{\pi}{6}}^{\frac{\pi}{4}} \frac{\sec(\theta)}{\tan(\theta)^2}\, d\theta = -\sqrt{2}+2$$

8. There are unique constants A and B with A an integer and B a rational number such that

$$\int_0^{\frac{\pi}{2}} \sin(x)^2 \cos(x)^6 \, dx = A + B \int_0^{\frac{\pi}{2}} \cos(x)^6 \, dx. \quad \text{What is B?}$$

a) $\frac{1}{2}$ **b)** $\frac{1}{3}$ **c)** $\frac{2}{3}$ **d)** $\frac{1}{4}$ **e)** $\frac{3}{4}$

f) $\frac{1}{6}$ **g)** $\frac{5}{6}$ **h)** $\frac{1}{8}$ **i)** $\frac{3}{8}$ **j)** $\frac{7}{8}$

Solution: h

$$\int \sin(x)^2 \cos(x)^6 \, dx = \int (1 - \cos(x)^2) \cos(x)^6 \, dx$$

$$\int \sin(x)^2 \cos(x)^6 \, dx = \int \cos(x)^6 \, dx - \int \cos(x)^8 \, dx$$

$$\int \sin(x)^2 \cos(x)^6 \, dx = \int \cos(x)^6 \, dx - \left(\frac{\sin(x)\cos(x)^7}{7} + \frac{7 \int \cos(x)^6 \, dx}{8} \right)$$

$$\int \sin(x)^2 \cos(x)^6 \, dx = -\frac{\sin(x)\cos(x)^7}{7} + \frac{1}{8} \int \cos(x)^6 \, dx$$

The expression on the rhs evaluates to 0 at both $x = 0$ and $x = \frac{\pi}{2}$.

Therefore,

$$\int_0^{\frac{\pi}{2}} \sin(x)^2 \cos(x)^6 \, dx = \text{A} + \text{B} \int_0^{\frac{\pi}{2}} \cos(x)^6 \, dx$$

with A = 0 and B = 1/8.

Verification using Maple's built-in integrator:

```
> int(sin(x)^2*cos(x)^6,x = 0 .. Pi/2);
  (1/8)*int( cos(x)^6,x = 0 .. Pi/2);
```

$$\frac{5\pi}{256}$$

$$\frac{5\pi}{256}$$

9. Make a trigonometric substitution to express the integral $\displaystyle\int \frac{1}{\sqrt{x^2-9}}\,dx$

in the form $A\displaystyle\int \sin(\theta)^p \cos(\theta)^q\, d\theta$**. Identify the entries of the ordered pair (p , q).**

a) (−1, −1) b) (−1, 0) c) (−1, 1) d) (−1, 2) e) (0, −1)
f) (0, 1) g) (1, −1) h) (1, 0) i) (1, 1) j) (1, 2)

Solution: e

The trigonometric substitution $x = 3\sec(\theta)$, $dx = 3\sec(\theta)\tan(\theta)\,d\theta$ transforms the given integral to

$$\int \frac{3\sec(\theta)\tan(\theta)}{\sqrt{9(\sec(\theta)^2-1)}}\,d\theta,$$

or

$$\int \frac{\sec(\theta)\tan(\theta)}{\sqrt{\tan(\theta)^2}}\,d\theta,$$

or

$$\int \sec(\theta)\,d\theta,$$

or $$\int \frac{1}{\cos(\theta)}\, d\theta.$$

The integrand is the 0^{th} power of $\sin(\theta)$ times the -1^{st} power of $\cos(\theta)$.

10. Make a trigonometric substitution to express the integral

$$\int \frac{1}{x\sqrt{4+x^2}}\, dx \quad \text{in the form } A \int \sin(\theta)^p \cos(\theta)^q \, d\theta.$$

Identify the entries of the ordered pair (p , q).

a) (−1, −1) b) (−1, 0) c) (−1, 1) d) (−1, 2) e) (0, −1)
f) (0, 1) g) (1, −1) h) (1, 0) i) (1, 1) j) (1, 2)

Solution: b

```
> eqn1 := Int(1/x/(4+x^2)^(1/2), x) = changevar(x=2*tan(theta),
  Int(1/x/(4+x^2)^(1/2), x), theta);
  #
  # Make the change of variable  x=2*tan(theta), dx =
  2*sec(theta)^2*dtheta
```

$$eqn1 := \int \frac{1}{x\sqrt{4+x^2}}\, dx = \int \frac{1+\tan(\theta)^2}{\tan(\theta)\sqrt{4+4\tan(\theta)^2}}\, d\theta$$

```
> eqn2 := lhs(eqn1) = subs( {4+4*tan(theta)^2 = 4*sec(theta)^2
  , 1+tan(theta)^2 = sec(theta)^2}, rhs(eqn1));
  #
  # Simplify: 1 +  tan(theta)^2 =  sec(theta)^2 and
  #           4 + 4tan(theta)^2 = 4sec(theta)^2
```

$$eqn2 := \int \frac{1}{x\sqrt{4+x^2}}\, dx = \int \frac{1}{4}\, \frac{\sec(\theta)^2\sqrt{4}}{\tan(\theta)\sqrt{\sec(\theta)^2}}\, d\theta$$

```
> eqn3 := lhs(eqn2) =
  (1/2)*Int(sec(theta)^2/(tan(theta)*sec(theta)),theta);
```

$$eqn3 := \int \frac{1}{x\sqrt{4+x^2}}\,dx = \frac{1}{2}\int \frac{\sec(\theta)}{\tan(\theta)}\,d\theta$$

```
> eqn4 := lhs(eqn3) = (1/2)*Int(
  convert(integrand(rhs(eqn3)),sincos), theta);
```

$$eqn4 := \int \frac{1}{x\sqrt{4+x^2}}\,dx = \frac{1}{2}\int \frac{1}{\sin(\theta)}\,d\theta$$

So, $A = \frac{1}{2}$, $p = -1$, $q = 0$.

11. Make a trigonometric substitution to express the integral

$$\int \frac{\sqrt{16-x^2}}{x}\,dx \quad \text{in the form } A\int \sin(\theta)^p \cos(\theta)^q\,d\theta.$$

Identify the entries of the ordered pair (p , q).

a) (–1, –1) b) (–1, 0) c) (–1, 1) d) (–1, 2) e) (0, –1)
f) (0, 1) g) (1, –1) h) (1, 0) i) (1, 1) j) (1, 2)

Solution: d

```
> eqn1 := Int( sqrt(16-x^2)/x, x) = changevar(x = 4*sin(theta),
  Int( sqrt(16-x^2)/x, x), theta);
  #
  # Make the substitution x = 4*sin(theta), dx =
  4*cos(theta)*dtheta
```

$$eqn1 := \int \frac{\sqrt{16-x^2}}{x}\,dx = \int \frac{\sqrt{16-16\sin(\theta)^2}\,\cos(\theta)}{\sin(\theta)}\,d\theta$$

```
> eqn2 := lhs(eqn1) = subs( 16-16*sin(theta)^2 =
  16*cos(theta)^2,  rhs(eqn1) );
```

$$eqn2 := \int \frac{\sqrt{16-x^2}}{x}\,dx = \int \frac{\sqrt{16}\,\sqrt{\cos(\theta)^2}\,\cos(\theta)}{\sin(\theta)}\,d\theta$$

So, $A = 4, p = -1, q = 2$.

12. Calculate the left side of the equation $$\int_2^5 \frac{5x+3}{(x-1)(x+7)}\,dx = A\ln(2) + B\ln(3),$$

expressing your evaluation in the form that appears on the right side of the equation. The values of A and B for which the equation holds are integers and they are unique. What is A?

a) 1	**b) 2**	**c) 3**	**d) 4**	**e) 5**
f) 6	**g) 7**	**h) 8**	**i) 9**	**j) 10**

Solution: j

```
> eqn := (5*x+3)/(x^2+6*x-7) = convert( (5*x+3)/(x^2+6*x-7),
  parfrac, x);
```

$$eqn := \frac{5x+3}{x^2+6x-7} = \frac{1}{x-1} + \frac{4}{x+7}$$

```
> Int(lhs(eqn),x=2..5) = int(rhs(eqn),x=2..5);
```

$$\int_2^5 \frac{5x+3}{x^2+6x-7}\,dx = 10\ln(2) - 4\ln(3)$$

Verification using Maple's builtin integrator

```
> Int((5*x+3)/(x-1)/(x+7),x = 2 .. 5) =
  int((5*x+3)/(x-1)/(x+7),x = 2 .. 5);
```

$$\int_2^5 \frac{5x+3}{(x-1)(x+7)}\,dx = 10\ln(2) - 4\ln(3)$$

13. Calculate the partial fraction decomposition of $\dfrac{x^3+2x^2+2x+3}{x^2(x^2+x+1)}$.

Write the decomposition in the usual way, each summand of the decomposition being a rational function with a denominator that is a factor of the denominator of the given expression: in particular, the coefficient of the highest power of x **in each denominator is 1. Of the coefficients of the polynomials that are the numerators of the summands, exactly two are positive. What is the sum of the two positive coefficients?**

a)	**5**	**b)**	**6**	**c)**	**7**	**d)**	**8**	**e)**	**9**
f)	**10**	**g)**	**11**	**h)**	**12**	**i)**	**13**	**j)**	**14**

Solution: a

Maple answers this question with a one-line command:

```
> convert( (x^3+2*x^2+2*x+3)/(x^2*(x^2+x+1)) , parfrac, x);
  #
  # Calculates the partial fraction decomposition of first
  argument
  # treating the third argument as the identity variable.
  # There are many conversions Maple performs, hence the need
  for the second argument, parfrac.
```

$$\frac{3}{x^2}+\frac{2x}{x^2+x+1}-\frac{1}{x}$$

Note: Maple has written $-\dfrac{1}{x}$ instead of $+\dfrac{-1}{x}$.

Here are the details of the partial fraction decomposition calculation:

```
> eqn1 := (x^3+2*x^2+2*x+3)/(x^2*(x^2+x+1)) = A/x + B/x^2 +
  (C*x+E)/(x^2 + x + 1);
  #
  # The form of the partial fraction expansion
```

$$eqn1 := \frac{x^3+2x^2+2x+3}{x^2(x^2+x+1)}=\frac{A}{x}+\frac{B}{x^2}+\frac{Cx+E}{x^2+x+1}$$

```
> eqn2 := lhs(eqn1) = normal(rhs(eqn1));
  #
  # Gets a common denominator on the right side
```

$$eqn2 := \frac{x^3+2x^2+2x+3}{x^2(x^2+x+1)}=\frac{Ax^3+Ax^2+Ax+Bx^2+Bx+B+x^3C+x^2E}{x^2(x^2+x+1)}$$

```
> eqn3 := numer(lhs(eqn2)) = numer(rhs(eqn2));
  #
  # Equates numerators of equal fractions, given that they have
  the same denominators
```

$$eqn3 := x^3 + 2\,x^2 + 2\,x + 3 = A\,x^3 + A\,x^2 + A\,x + B\,x^2 + B\,x + B + x^3\,C + x^2\,E$$

```
> eqn4 := subs(x = 0, eqn3);
  #
  # The original denominator has one real root, 0. May as well
  substitute it to pick off one coefficient immediately
```

$$eqn4 := 3 = B$$

```
> eqn5 := subs(B = 3, eqn3);
  #
  # Now that the value of B is known, substitute it back into
  the main identity, eqn3
```

$$eqn5 := x^3 + 2\,x^2 + 2\,x + 3 = A\,x^3 + A\,x^2 + A\,x + 3\,x^2 + 3\,x + 3 + x^3\,C + x^2\,E$$

```
> eqn6 := lhs(eqn5) = collect(rhs(eqn5), x);
  #
  # Expand and collect terms on the right side of eqn5
```

$$eqn6 := x^3 + 2\,x^2 + 2\,x + 3 = (A + C)\,x^3 + (3 + A + E)\,x^2 + (3 + A)\,x + 3$$

```
> solve( {A+C = 1, (3+A+E) = 2, 3+A = 2}, {A,C,E} );
  #
  # Equate coefficients of like powers of x on both sides of
  eqn6. Then solve the simultaneous equations.
```

$$\{C = 2, E = 0, A = -1\}$$

14. Calculate $\displaystyle\int_{-1}^{0} \frac{8\,x + 16}{(x^2 + 2\,x + 2)^2}\,dx.$

a) $2\,\pi + 1$ **b)** $2\,(\pi + 1)$ **c)** $\pi + 2$ **d)** $4\,(\pi - 1)$ **e)** $\pi + 3$
f) $3\,\pi - 2$ **g)** $\pi + 4$ **h)** $2\,\pi - 1$ **i)** $2\,\pi + 3$ **j)** $4\,\pi - 5$

Solution: g

```
> Int((8*x+16)/((x^2+2*x+2)^2),x = -1 .. 0)  =
  int((8*x+16)/((x^2+2*x+2)^2),x = -1 .. 0);
```

$$\int_{-1}^{0} \frac{8x+16}{(x^2+2x+2)^2}\,dx = \pi + 4$$

```
> eqn1 := Int((8*x+16)/(x^2+2*x+2)^2,x) =
  Int((8*x+16)/completesquare(x^2+2*x+2,x)^2,x);
```

$$eqn1 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = \int \frac{8x+16}{((x+1)^2+1)^2}\,dx$$

```
> eqn2 := u = x + 1;
```

$$eqn2 := u = x + 1$$

```
> eqn3 := lhs(eqn1) = changevar(eqn2, rhs(eqn1), u);
```

$$eqn3 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = \int \frac{8+8u}{(u^2+1)^2}\,du$$

```
> eqn4 := lhs(eqn3) = expand( rhs(eqn3) );
```

$$eqn4 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = 8\int \frac{1}{(u^2+1)^2}\,du + 8\int \frac{u}{(u^2+1)^2}\,du$$

```
> eqn5 := lhs(eqn4) = 8*Int(cos(theta)^2, theta) +
  8*changevar(w=u^2+1, Int(1/(u^2+1)^2*u,u), w);
  #
  # The first integral on the right side is treated with the
  substitution u = tan(theta)
  # The second integral on the right is treated with the
  substitution w = u^2+1
```

$$eqn5 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = 8\int \cos(\theta)^2\,d\theta + 8\int \frac{1}{2w^2}\,dw$$

```
> eqn6 := lhs(eqn5) = value(rhs(eqn5));
```

$$eqn6 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = 4\cos(\theta)\sin(\theta) + 4\theta - \frac{4}{w}$$

```
> eqn7 := lhs(eqn6) = subs( {theta = arctan(u), w = u^2+1},
  rhs(eqn6) );
```

$$eqn7 := \int \frac{8x+16}{(x^2+2x+2)^2}\,dx = 4\cos(\arctan(u))\sin(\arctan(u)) + 4\arctan(u) - \frac{4}{u^2+1}$$

```
> eqn8 := lhs(eqn7) = simplify( rhs(eqn7) );
  #
```

```
# Draw a right triangle with angle theta and opposite side u,
adjacent side 1.
# The triangle shows that cos(arctan(u)) = cos(theta) =
1/sqrt(u^2+1), and
#                          sin(arctan(u)) = sin(theta) =
u/sqrt(u^2+1)
```

$$eqn8 := \int \frac{8\,x+16}{(x^2+2\,x+2)^2}\,dx = \frac{4\,(u+\arctan(u)\,u^2+\arctan(u)-1)}{u^2+1}$$

```
> eqn9 := lhs(eqn8) = subs(eqn2, rhs(eqn8));
#
# Resubstitute to express antiderivative in terms of x
```

$$eqn9 := \int \frac{8\,x+16}{(x^2+2\,x+2)^2}\,dx = \frac{4\,(x+\arctan(x+1)\,(x+1)^2+\arctan(x+1))}{(x+1)^2+1}$$

```
> eqn10 := Int((8*x+16)/(x^2+2*x+2)^2,x = -1 .. 0) = subs(x=0,
rhs(eqn9)) - subs(x=-1, rhs(eqn9));
```

$$eqn10 := \int_{-1}^{0} \frac{8\,x+16}{(x^2+2\,x+2)^2}\,dx = 4\arctan(1)+4-4\arctan(0)$$

```
> eqn11 := lhs(eqn10) = simplify(rhs(eqn10));
```

$$eqn11 := \int_{-1}^{0} \frac{8\,x+16}{(x^2+2\,x+2)^2}\,dx = \pi+4$$

Verification using Maple's builtin integrator:

```
> Int((8*x+16)/((x^2+2*x+2)^2),x = -1 .. 0) =
int((8*x+16)/((x^2+2*x+2)^2),x = -1 .. 0);
```

$$\int_{-1}^{0} \frac{8\,x+16}{(x^2+2\,x+2)^2}\,dx = \pi+4$$

15. Evaluate the convergent improper integral $\displaystyle\int_{2}^{\infty} \frac{2\ln(x)}{x^2}\,dx$.

a) $\ln(2)+1$
b) $\ln(2)+2$
c) $2\ln(2)+1$
d) $2(\ln(2)+1)$
e) $4(\ln(2)+1)$
f) $2(2\ln(2)+1)$
g) $4\ln(2)+1$
h) $8(\ln(2)+1)$
i) $4(2\ln(2)+1)$
j) $2(4\ln(2)+1)$

Solution: a

First, the antiderivative of the integrand. We use integration by parts with $u=\ln(x)$ and $dv=\frac{2\,dx}{x^2}$.

```
> eqn1 := Int(2*ln(x)/x^2,x) = intparts(Int(2*ln(x)/x^2,x),
  ln(x));
  #
  # Integration by parts with u = ln(x) and dv = (2/x^2)*dx
```

$$eqn1 := \int \frac{2\ln(x)}{x^2}\,dx = -\frac{2\ln(x)}{x} - \int -\frac{2}{x^2}\,dx$$

```
> eqn2 := lhs(eqn1) = value(rhs(eqn1));
```

$$eqn2 := \int \frac{2\ln(x)}{x^2}\,dx = -\frac{2\ln(x)}{x} - \frac{2}{x}$$

As a result,

```
> eqn3 := Int(2*ln(x)/x^2,x = 2..N) = subs(x=N, rhs(eqn2)) -
  subs(x=2, rhs(eqn2));
```

$$eqn3 := \int_2^N \frac{2\ln(x)}{x^2}\,dx = -\frac{2\ln(N)}{N} - \frac{2}{N} + \ln(2) + 1$$

Now for the improper integral:

```
> eqn4 := Int(2*ln(x)/x^2,x=2..infinity) =
  Limit(Int(2*ln(x)/x^2,x=2..N), N = infinity);
```

$$eqn4 := \int_2^{\infty} \frac{2\ln(x)}{x^2}\,dx = \lim_{N\to\infty} \int_2^{N} \frac{2\ln(x)}{x^2}\,dx$$

```
> eqn5 := lhs(eqn4) = Limit( rhs(eqn3), N = infinity);
```

$$eqn5 := \int_2^{\infty} \frac{2\ln(x)}{x^2}\,dx = \lim_{N\to\infty} \left(-\frac{2\ln(N)}{N} - \frac{2}{N} + \ln(2) + 1\right)$$

```
> eqn6 := lhs(eqn5) = value(rhs(eqn5));
  #
  # The two summands involving N have 0 as their limits. This
  is obvious
  # for the second summand 2/N. For ln(N)/N, the limit is
  calculated using L'Hopital's Rule
```

$$eqn6 := \int_2^{\infty} \frac{2\ln(x)}{x^2}\,dx = \ln(2) + 1$$

16. Evaluate the convergent improper integral $\displaystyle\int_0^4 \frac{3x}{\sqrt{4-x}}\,dx$.

a) 16	b) 20	c) 24	d) 28	e) 32
f) 36	g) 40	h) 44	i) 48	j) 52

Solution: e

```
> eqn1 := Int(3*x/sqrt(4-x),x = 0..4) =
  Limit(Int(3*x/sqrt(4-x),x = 0..4-epsilon), epsilon=0,right);
```

$$eqn1 := \int_0^{4} \frac{3x}{\sqrt{4-x}}\,dx = \lim_{\varepsilon\to 0+} \int_0^{4-\varepsilon} \frac{3x}{\sqrt{4-x}}\,dx$$

```
> eqn2 := lhs(eqn1) = Limit(changevar(u = 4-x,
  Int(3*x/sqrt(4-x),x = 0..4-epsilon), u), epsilon=0, right);
```

$$eqn2 := \int_0^4 \frac{3\,x}{\sqrt{4-x}}\,dx = \lim_{\varepsilon \to 0+} \int_4^{\varepsilon} -\frac{3\,(4-u)}{\sqrt{u}}\,du$$

```
> eqn3 := lhs(eqn2) = Limit( map(expand, Int(-3*(4-u)/u^(1/2),u
  = 4 .. epsilon)) , epsilon=0, right);
```

$$eqn3 := \int_0^4 \frac{3\,x}{\sqrt{4-x}}\,dx = \lim_{\varepsilon \to 0+} \int_4^{\varepsilon} -\frac{12}{\sqrt{u}} + 3\sqrt{u}\,du$$

```
> eqn4 := lhs(eqn3) = Limit(  value(Int(-12/u^(1/2)+3*u^(1/2),u
  = 4 .. epsilon)) , epsilon=0, right);
```

$$eqn4 := \int_0^4 \frac{3\,x}{\sqrt{4-x}}\,dx = \lim_{\varepsilon \to 0+} -24\sqrt{\varepsilon} + 2\,\varepsilon^{(3/2)} + 32$$

```
> eqn5 := lhs(eqn4) = value(rhs(eqn4));
```

$$eqn5 := \int_0^4 \frac{3\,x}{\sqrt{4-x}}\,dx = 32$$

Using Maple's builtin integrator more directly:

```
> Int(3*x/sqrt(4-x),x=0..4) = int(3*x/sqrt(4-x),x=0..4);
```

$$\int_0^4 \frac{3\,x}{\sqrt{4-x}}\,dx = 32$$

17. In 1998 the U.S. Lorenz function L for household income had the following observed values:

L(0) = 0, L(20) = 3.6, L(40) = 12.6, L(60) = 27.6,   L(80) = 50.8, L(95) = 78.6,   L(100) = 100

(Familarity with the Lorenz function is not needed for this problem, nor is it even helpful.)

Using *all* the given observations in conjunction with trapezoids as the approximation method, estimate

$$\int_0^{100} \mathrm{L}(x)\,dx.$$

a) 2793 b) 2794 c) 2795 d) 2796 e) 2797
f) 2798 g) 2799 h) 2800 i) 2801 j) 2802

Solution: i

The Trapezoidal Rule formula

$$\int_{x_0}^{x_N} L(x)\,dx \quad \text{is approximately equal to} \quad \frac{\Delta x}{2} \Big(L(x_0) + 2\,L(x_1) + 2\,L(x_2) + \ldots + 2\,L(x_{N-1}) + L(x_N) \Big)$$

is valid only when the nodes $x_0, x_1, x_2, \ldots, x_{N-1}, x_N$ are equally spaced. We therefore decompose the
integral as

$$\int_0^{80} L(x)\,dx + \int_{80}^{95} L(x)\,dx + \int_{95}^{100} L(x)\,dx$$

and use the Trapezoidal Rule formula for the integral over the interval [0,80]. The five nodes are
$x_0 = 0$, $x_1 = 20$, $x_2 = 40$, $x_3 = 60$, $x_4 = 80$, and $\Delta x = 20$.The remaining
two integrals are estimated using one trapezoid each. Our estimate is

```
> (20/2)*(1*0 + 2*3.6 + 2*12.6 + 2*27.6 + 1*50.8) +
  (15/2)*(50.8 + 78.6) + (5/2)*(78.6 + 100);
```

2801.000000

18. A quadratic polynomial $f(x) = A\,x^2 + B\,x + C$ has the following values: $f(-1) = -20$, $f(2) = -2$, $f(5) = 34$. Calculate

$$\int_{-1}^{5} f(x)\,dx.$$

a) 5 b) 6 c) 7 d) 8 e) 9
f) 10 g) 11 h) 12 i) 13 j) 14

Solution: b

According to Simpson's Rule with $N = 2$ subintervals and $\Delta x = \dfrac{5-(-1)}{2} = 3$, the exact value of the given integral is

$$\int_{-1}^{5} \mathbf{f}(x)\, dx = \frac{\Delta x}{3}\left(\mathrm{f}(-1) + 4\,\mathrm{f}(2) + \mathrm{f}(5) \right) = \frac{3}{3}\left(-20 + 4\,(-2) + 34 \right) = 6.$$

N.B. Simpson's Rule is generally an approximation, but, by design, it is exact for quadratic functions.

An unpleasant alternative solution is to use the three given equations to solve for the three unspecified
constants A, B, and C. Then the integral can be routinely evaluated.

```
> f := x -> A*x^2 + B*x + C;
```

$$f := x \to A\,x^2 + B\,x + C$$

```
> eqn1 := f(-1) = -20;
  eqn2 := f(2) = -2;
  eqn3 := f(5) = 34;
```

$$eqn1 := A - B + C = -20$$

$$eqn2 := 4\,A + 2\,B + C = -2$$

$$eqn3 := 25\,A + 5\,B + C = 34$$

```
> soln_set := solve( {eqn1, eqn2, eqn3}, {A,B,C} );
```

$$soln\_set := \{ A = 1,\, B = 5,\, C = -16 \}$$

```
> f := x -> subs(soln_set, A*x^2 + B*x + C);
```

$$f := x \to \mathrm{subs}(soln\_set,\, A\,x^2 + B\,x + C)$$

```
> f(x);
```

$$x^2 + 5\,x - 16$$

```
> Int(f(x), x = -1 .. 5) = int(f(x), x = -1 .. 5);
```

$$\int_{-1}^{5} x^2 + 5\,x - 16\, dx = 6$$

19. Calculate the length of the graph of $y = \ln(\sec(x))$ from the origin to the point $(\pi/3,\, \ln(2))$.

a) $2\ln(2)$ **b)** $2\ln(3)$ **c)** $\ln(1+\sqrt{2})$ **d)** $\ln(1+\sqrt{3})$ **e)** $\ln(2+\sqrt{2})$

f) $\ln(2+\sqrt{3})$ g) $\ln(3+\sqrt{2})$ h) $\ln(3+\sqrt{3})$ i) $2\ln(1+\sqrt{2})$ j) $2\ln(1+\sqrt{3})$

Solution: f

For the function

```
> f := x -> ln(sec(x));
```

$$f := x \to \ln(\sec(x))$$

we have

```
> Diff( f(x), x)^2 = diff( f(x), x)^2;
```

$$\left(\frac{d}{dx}\ln(\sec(x))\right)^2 = \tan(x)^2$$

and therefore

```
> 1 + Diff( f(x), x)^2 = sec(x)^2;
```

$$1+\left(\frac{d}{dx}\ln(\sec(x))\right)^2 = \sec(x)^2$$

and

```
> sqrt( 1 + Diff( f(x), x)^2 ) = sec(x);
```

$$\sqrt{1+\left(\frac{d}{dx}\ln(\sec(x))\right)^2} = \sec(x)$$

Because

```
> Int(sec(x),x) = int(sec(x),x);
```

$$\int \sec(x)\,dx = \ln(\sec(x)+\tan(x))$$

the arc length L is

```
> L := subs(x = Pi/3, ln(sec(x)+tan(x))) - subs(x = 0,
  ln(sec(x)+tan(x)));
```

$$L := \ln\left(\sec\left(\frac{\pi}{3}\right) + \tan\left(\frac{\pi}{3}\right)\right) - \ln(\sec(0) + \tan(0))$$

which simplifies to

```
> simplify(L);
```

$$\ln(2 + \sqrt{3})$$

Using Maple's builtin integrator to verify,

```
> Int( sqrt(1 + Diff( ln(sec(x)),x)^2),x=0..Pi/3) = int( sqrt(1
  + diff( ln(sec(x)),x)^2),x=0..Pi/3);
```

$$\int_0^{\frac{\pi}{3}} \sqrt{1 + \left(\frac{d}{dx}\ln(\sec(x))\right)^2}\, dx = \ln(2 + \sqrt{3})$$

20. The graph of $y = 2\sqrt{x}$ from $x = 0$ to $x = 3$ is rotated about the x-axis. What is the area of the resulting surface?

a) $\frac{52\pi}{3}$ b) $\frac{35\pi}{2}$ c) $\frac{53\pi}{3}$ d) 18π e) $\frac{73\pi}{4}$
f) $\frac{37\pi}{2}$ g) $\frac{56\pi}{3}$ h) $\frac{75\pi}{4}$ i) 19π j) $\frac{58\pi}{3}$

Solution: g

For the function

```
> f := x -> 2*sqrt(x);
```

$$f := x \to 2\sqrt{x}$$

we have

```
> Diff( f(x), x)^2 = diff( f(x), x)^2;
```

$$\left(\frac{d}{dx}(2\sqrt{x})\right)^2 = \frac{1}{x}$$

and therefore

```
> 1 + Diff( 'f(x)', x)^2 = normal(1 + diff( f(x), x)^2);
```

$$1+\left(\frac{d}{dx}\,\mathrm{f}(x)\right)^2=\frac{x+1}{x}$$

and

```
> 'f(x)'*sqrt( 1 + Diff( 'f(x)', x)^2 ) = f(x)*sqrt(normal(1 +
  diff( f(x), x)^2));
```

$$\mathrm{f}(x)\sqrt{1+\left(\frac{d}{dx}\,\mathrm{f}(x)\right)^2}=2\sqrt{x}\sqrt{\frac{x+1}{x}}$$

or

```
> 'f(x)'*sqrt( 1 + Diff( 'f(x)', x)^2 ) = 2*sqrt(x+1);
```

$$\mathrm{f}(x)\sqrt{1+\left(\frac{d}{dx}\,\mathrm{f}(x)\right)^2}=2\sqrt{x+1}$$

This results in an integral that is calculated by the simple substitution $u=x+1$, $du=dx$. Because

```
> Int(2*Pi*'f(x)'*sqrt( 1 + Diff( 'f(x)', x)^2 ), x = 0 .. 3) =
  2*Pi*int(2*sqrt(x+1), x=0..3);
```

$$\int_0^3 2\,\pi\,\mathrm{f}(x)\sqrt{1+\left(\frac{d}{dx}\,\mathrm{f}(x)\right)^2}\,dx=\frac{56\,\pi}{3}$$

Using Maple's builtin integrator to verify,

```
> 2*Pi*Int( f(x)*sqrt( 1 + Diff( f(x) , x)^2 ), x = 0..3) =
  2*Pi*int( f(x)*sqrt( 1 + diff( f(x) , x)^2 ), x = 0..3);
```

$$2\,\pi\int_0^3 2\sqrt{x}\sqrt{1+\left(\frac{d}{dx}\,(2\sqrt{x})\right)^2}\,dx=\frac{56\,\pi}{3}$$

Code